УРОК №2
ТЕМА «Алгоритмы сортировки порядка n*log2(n)»

Цель урока:
рассмотреть алгоритмы сортировки линейного массива порядка n*log2(n).
Задачи урока:

образовательные:
изучить алгоритмы сортировки линейного массива порядка n*log2(n); реализовать эти алгоритмы на языке программирования; научиться применять эти алгоритмы при решении задач;
развивающие:
развитие алгоритмического мышления, способностей к формализации, элементов системного мышления;

воспитательные:
воспитание чувства ответственности за результаты своего труда.
Материалы и оборудование к уроку:
ПК.
Тип урока:

комбинированный.

Форма проведения урока:

беседа.
План урока:
1. Квадратичные алгоритмы сортировки
2. Алгоритмы порядка n log2 n.
3. Задачи.


Ход урока:

1. Квадратичные алгоритмы сортировки
Предложить алгоритм сортировки произвольного линейного целочисленного массива. Т.е. Пусть a[1], ..., a[n] - целые числа. Требуется переставить некоторые элементы местами так, чтобы во вновь полученном массиве выполнялось: a[1]<=...<=a[n].
for i:=1 to n-1 do
for j:=1 to n-i do

if a[j]<a[j+1] then begin поменять местами a[j] и a[j+1] end;
или
k:=1;

{первые k элементов упорядочены}

while k <> n do begin

t := k+1;

{k+1-ый элемент продвигается к началу, пока не займет

надлежащего места, t - его текущий номер}

while (t > 1) and (x[t] < x[t-1]) do 
           begin

           поменять x[t-1] и x[t];

           t := t - 1;

           end;
k:=k+1;
end;
или
k := 0;

{k наименьших элементов массива x установлены на свои места}

while k <> n do 
   begin

   s := k + 1; t := k + 1;

   {x[s] - наименьший среди x[k+1]...x[t] }

   while t<>n do 
            begin

            t := t + 1;

            if x[t] < x[s] then 
                           begin

                           s := t;

                           end;

            end;

{x[s] - наименьший среди x[k+1]..x[n] }

... переставить x[s] и x[k+1];

k := k + 1;

end;
Предложенные решения требуют числа действий, пропорционального n*n. Существуют более эффективные алгоритмы.
2. Алгоритмы порядка n log2 n.
Предложить алгоритм сортировки, число действий которого было бы порядка nlog2n, то есть не превосходило бы C*n*log2(n) для некоторого C и для всех n.

Решение 1. (сортировка слиянием).

Пусть k - положительное целое число. Разобьем массив x[1]..x[n] на отрезки длины k. (Первый - x[1]..x[k], затем x[k+1]..x[2k] и т.д.) Последний отрезок будет неполным, если n не делится на k. Назовем массив k-упорядоченным, если каждый из этих отрезков в отдельности упорядочен. Любой массив 1-упорядочен. Если массив k-упорядочен и n<=k, то он упорядочен.

Мы опишем, как преобразовать k-упорядоченный массив в 2k-упорядоченный (из тех же элементов). С помощью этого преобразования алгоритм записывается так:

  k:=1;

  {массив x является k-упорядоченным}

  while k < n do begin

  | .. преобразовать k-упорядоченный массив в 2k-упорядоченный;

  | k := 2 * k;

  end;

Требуемое преобразование состоит в том, что мы многократно "сливаем" два упорядоченных отрезка длины не больше k в один упорядоченный отрезок. Пусть процедура слияние (p,q,r: integer) при p <=q <= r сливает отрезки x[p+1]..x[q] и x[q+1]..x[r] в упорядоченный отрезок x[p+1]..x[r] (не затрагивая других частей массива x).

Тогда преобразование k-упорядоченного массива в 2k-упорядоченный осуществляется так:

  t:=0;

  {t кратно 2k или t = n, x[1]..x[t] является

   2k-упорядоченным; остаток массива x не изменился}

  while t + k < n do begin

  | p := t;

  | q := t+k;

  | ...r := min (t+2*k, n); {min(a,b) - минимум из a и b}

  | слияние (p,q,r);

  | t := r;

  end;

Слияние требует вспомогательного массива для записи результатов слияния - обозначим его b. Через p0 и q0 обозначим номера последних элементов участков, подвергшихся слиянию, s0 - последний записанный в массив b элемент. На каждом шаге слияния производится одно из двух действий:

        b[s0+1]:=x[p0+1];

        p0:=p0+1;

        s0:=s0+1;

или
        b[s0+1]:=x[q0+1];

        q0:=q0+1;

        s0:=s0+1;

Первое действие (взятие элемента из первого отрезка) может производиться при одновременном выполнении двух условий:

(1) первый отрезок не кончился (p0 < q);

(2) второй отрезок кончился (q0 = r) или не кончился, но элемент в нем не меньше очередного элемента второго отрезка [(q0 < r) и (x[p0+1] <= x[q0+1])].

Аналогично для второго действия. Итак, получаем

  p0 := p; q0 := q; s0 := p;

  while (p0 <> q) or (q0 <> r) do begin

  | if (p0 < q) and ((q0 = r) or ((q0 < r) and

  | |                (x[p0+1] <= x[q0+1]))) then begin

  | | b [s0+1] := x [p0+1];

  | | p0 := p0+1;

  | | s0 := s0+1;

  | end else begin

  | | {(q0 < r) and ((p0 = q) or ((p0<q) and

  | |   (x[p0+1] >= x[q0+1])))}

  | | b [s0+1] := x [q0+1];

  | | q0 := q0 + 1;

  | | s0 := s0 + 1;

  | end;

  end;

(Если оба отрезка не кончены и первые невыбранные элементы в них равны, то допустимы оба действия; в программе выбрано первое.)

Остается лишь переписать результат слияния обратно в x. (Предупреждение. Если обратное копирование выполняется вне процедуры слияния, то не забудьте переписать последний отрезок.)

Сортировка слиянием хороша тем, что она не требует, чтобы весь сортируемый массив помещался в оперативной памяти. Можно сначала отсортировать такие куски, которые помещаются в памяти (например, с помощью дерева), а затем сливать полученные файлы.

Программа может выглядеть так:
{$A+,B-,D+,E-,F-,G-,I+,L+,N+,O-,P-,Q-,R-,S+,T-,V+,X+}

{$M 65520,0,655360}

type mass=array[1..32000] of byte;

var x:mass;

    i,n:longint;

    f:text;

procedure sort(var x:mass);

var k,t,p,q,r,p0,q0,s0:longint;

    b:mass;

    function min(x,y:longint):longint;

    begin

    if x<y then min:=x else min:=y;

    end;

begin

k:=1;

{массив x является k-упорядоченным}

while k < n do begin

{преобразовать k-упорядоченный массив в 2k-упорядоченный;}

t:=0;

{t кратно 2k или t = n, x[1]..x[t] является 2k-упорядоченным;

остаток массива x не изменился}

   while t + k < n do begin

   p := t;

   q := t+k;

   r := min (t+2*k, n); {min(a,b) - минимум из a и b}

        {слияние (p,q,r);}

        p0 := p; q0 := q; s0 := p;

        while (p0 <> q) or (q0 <> r) do begin

        if (p0 < q) and ((q0 = r) or ((q0 < r) and (x[p0+1] <= x[q0+1])))

                        then

                                begin

                                b [s0+1] := x [p0+1];

                                p0 := p0+1;

                                s0 := s0+1;

                                end

                        else begin

        {(q0 < r) and ((p0 = q) or ((p0<q) and(x[p0+1] >= x[q0+1])))}

                                b [s0+1] := x [q0+1];

                                q0 := q0 + 1;

                                s0 := s0 + 1;

                                end;

        end;

        for i:=p+1 to r do x[i]:=b[i];

    t := r;

    end;

k := 2 * k;

end;

end;

begin

randomize;

assign(f,'output.txt');rewrite(f);

Write('N=');readln(n);

for i:= 1 to n do x[i]:=random(100);

sort(x);

for i:=1 to n do write(f,x[i]:3);

close(f);

end.
Решение 2 (сортировка деревом).

Нарисуем "полное двоичное дерево" - картинку, в которой снизу один кружок, из него выходят стрелки в два других, из каждого - в два других и так далее:


[image: image1]
Будем говорить, что стрелки ведут "от отцов к сыновьям": у каждого кружка два сына и один отец (если кружок не в самом верху или низу). Предположим для простоты, что количество подлежащих сортировке чисел есть степень двойки, и они могут заполнить один из рядов целиком. Запишем их туда. Затем заполним часть дерева под ними по правилу:

число в кружке = минимум из чисел в кружках-сыновьях

Тем самым в корне дерева (нижнем кружке) будет записано минимальное число во всем массиве.

Изымем из сортируемого массива минимальный элемент. Для этого его надо вначале найти. Это можно сделать, идя от корня: от отца переходим к тому сыну, где записано то же число. Изъяв минимальный элемент, заменим его символом "бесконечность" и скорректируем более низкие ярусы (для этого надо снова пройти путь к корню). При этом считаем, что минимум из n и бесконечности равен n. Тогда в корне появится второй по величине элемент, мы изымаем его, заменяя бесконечностью и корректируя дерево. Так постепенно мы изымем все элементы в порядке возрастания, пока в корне не останется бесконечность.

При записи этого алгоритма полезно нумеровать кружки числами 1, 2, ... - при этом сыновьями кружка номер n являются кружки 2*n и 2*n+1. Подробное изложение этого алгоритма мы опустим, поскольку мы изложим более эффективный вариант, не требующий дополнительной памяти, кроме конечного числа переменных (в дополнение к сортируемому массиву).

Мы будем записывать сортируемые числа во всех вершинах дерева, а не только на верхнем уровне. Пусть x[1]..x[n] - массив, подлежащий сортировке. Вершинами дерева будут числа от 1 до n; о числе x[i] мы будем говорить как о числе, стоящем в вершине i. В процессе сортировки количество вершин дерева будет сокращаться.

Число вершин текущего дерева будем хранить в переменной k. Таким образом, в процессе работы алгоритма массив x[1]..x[n] делится на две части: в x[1]..x[k] хранятся числа на дереве, а в x[k+1].. x[n] хранится уже отсортированная в порядке возрастания часть массива - элементы, уже занявшие свое законное место.

На каждом шаге алгоритм будет изымать максимальный элемент дерева и помещать его в отсортированную часть, на освободившееся в результате сокращения дерева место.

Договоримся о терминологии. Вершинами дерева считаются числа от 1 до текущего значения переменной k. У каждой вершины s могут быть сыновья 2s и 2s+1. Если оба этих числа больше k, то сыновей нет; такая вершина называется листом. Если 2s=k, то вершина s имеет ровно одного сына (2s).

Для каждого s из 1..k рассмотрим "поддерево" с корнем в s: оно содержит вершину s и всех ее потомков (сыновей, внуков и т.д. - до тех пор, пока мы не выйдем из отрезка 1..k). Вершину s будем называть регулярной, если стоящее в ней число - максимальный элемент s-поддерева; s-поддерево назовем регулярным, если все его вершины регулярны. (В частности, любой лист образует регулярное одноэлементное поддерево.) Заметим, что истинность утверждения "s-поддерево регулярно" зависит не только от s, но и от текущего значения k.

Схема алгоритма такова:

  k:= n
  ... Сделать 1-поддерево регулярным;

  {x[1],..,x[k] <= x[k+1] <= ... <= x[n]; 1-поддерево регулярно,

   в частности, x[1] - максимальный элемент среди x[1]..x[k]}

  while k <> 1 do begin
  | ... обменять местами x[1] и x[k];

  | k := k - 1;

  | {x[1]..x[k-1] <= x[k] <=...<= x[n]; 1-поддерево регу-

  |   лярно везде, кроме, возможно, самого корня }

  | ... восстановить регулярность 1-поддерева всюду

  end;

В качестве вспомогательной процедуры нам понадобится процедура восстановления регулярности s-поддерева в корне. Вот она:

  {s-поддерево регулярно везде, кроме, возможно, корня}

  t := s;

  {s-поддерево регулярно везде, кроме, возможно, вершины t}

  while ((2*t+1 <= k) and (x[2*t+1] > x[t])) or

  |     ((2*t <= k) and (x[2*t] > x[t])) do begin

  | if (2*t+1 <= k) and (x[2*t+1] >= x[2*t]) then begin

  | | ... обменять x[t] и x[2*t+1];

  | | t := 2*t + 1;

  | end else begin

  | | ... обменять x[t] и x[2*t];

  | | t := 2*t;

  | end;

  end;

Чтобы убедиться в правильности этой процедуры, посмотрим на нее повнимательнее. Пусть в s-поддереве все вершины, кроме разве что вершины t, регулярны. Рассмотрим сыновей вершины t. Они регулярны, и потому содержат наибольшие числа в своих поддеревьях.

Таким образом, на роль наибольшего числа в t-поддереве могут претендовать число в самой вершине t и числа в ее сыновьях. (В первом случае вершина t регулярна, и все в порядке.) В этих терминах цикл можно записать так:

  while наибольшее число не в t, а в одном из сыновей do begin
  | if оно в правом сыне then begin
  | | поменять t с ее правым сыном; t:= правый сын

  | end else begin {наибольшее число - в левом сыне}

  | | поменять t с ее левым сыном; t:= левый сын

  | end
  end
После обмена вершина t становится регулярной (в нее попадает максимальное число t-поддерева). Не принявший участия в обмене сын остается регулярным, а принявший участие может и не быть регулярным. В остальных вершинах s-поддерева не изменились ни числа, ни поддеревья их потомков (разве что два элемента поддерева переставились), так что регулярность не нарушилась.

Эта же процедура может использоваться для того, чтобы сделать 1-поддерево регулярным на начальной стадии сортировки:

  k := n;

  u := n;

  {все s-поддеревья с s>u регулярны }

  while u<>0 do begin
  | {u-поддерево регулярно везде, кроме разве что корня}

  | ... восстановить регулярность u-поддерева в корне;

  | u:=u-1;

  end;

Теперь запишем процедуру сортировки на паскале (предполагая, что n - константа, x имеет тип arr = array [1..n] of integer).

  procedure sort (var x: arr);

  | var u, k: integer;

  | procedure exchange(i, j: integer);

  | | var tmp: integer;

  | | begin

  | | tmp  := x[i];

  | | x[i] := x[j];

  | | x[j] := tmp;

  | end;

  | procedure restore (s: integer);

  | | var t: integer;

  | | begin

  | | t:=s;

  | | while ((2*t+1 <= k) and (x[2*t+1] > x[t])) or

  | | |     ((2*t <= k) and (x[2*t] > x[t])) do begin

  | | | if (2*t+1 <= k) and (x[2*t+1] >= x[2*t]) then begin

  | | | | exchange (t, 2*t+1);

  | | | | t := 2*t+1;

  | | | end else begin

  | | | | exchange (t, 2*t);

  | | | | t := 2*t;

  | | | end;

  | | end;

  | end;

  begin

  | k:=n;

  | u:=n;

  | while u <> 0 do begin

  | | restore (u);

  | | u := u - 1;

  | end;

  | while k <> 1 do begin

  | | exchange (1, k);

  | | k := k - 1;

  | | restore (1);

  | end;

  end;

Программа может выглядеть так:

{$A+,B-,D+,E-,F-,G-,I+,L+,N+,O-,P-,Q-,R-,S+,T-,V+,X+}

{$M 16384,0,655360}

type mass=array[1..32000] of byte;

var x:mass;

    i,u,k,n,t:longint;

    f:text;

procedure swap(var x,y:byte);

var temp:byte;

begin

temp:=x;x:=y;y:=temp;

end;

procedure sreg(s,k:integer; var x:mass);

{s-поддерево регулярно везде, кроме, возможно, корня}

var t:integer;

begin

  t := s;

  {s-поддерево регулярно везде, кроме, возможно, вершины t}

  while ((t<=16001) and (2*t+1 <= k) and (x[2*t+1] > x[t])) or

        ((t<=16001) and (2*t <= k) and (x[2*t] > x[t])) do

        if (2*t+1 <= k) and (x[2*t+1] >= x[2*t]) then begin

                                                      swap(x[t],x[2*t+1]);

                                                      t := 2*t + 1;

                                                      end

                                                  else begin

                                                       swap(x[t],x[2*t]);

                                                       t := 2*t;

                                                       end;

end;

begin

randomize;

assign(f,'output.txt');rewrite(f);

Write('N=');readln(n);

for i:= 1 to n do x[i]:=random(256);

     {Сделать 1-поддерево регулярным}

{1}  k := n;

{1}  u := n;

{1}  {все s-поддеревья с s>u регулярны }

{1}  while u<>0 do begin

{1}  {u-поддерево регулярно везде, кроме разве что корня}

{1}  sreg(u,k,x);{восстановить регулярность u-поддерева в корне}

{1}  u:=u-1;

{1}  end;

     {x[1],..,x[k] <= x[k+1] <= ... <= x[n]; 1-поддерево регулярно,

     в частности, x[1] - максимальный элемент среди x[1]..x[k]}

{2}  while k <> 1 do begin

{2}  swap(x[1],x[k]);

{2}  k := k - 1;

{2}  {x[1]..x[k-1] <= x[k] <=...<= x[n]; 1-поддерево регулярно везде,

     кроме, возможно, самого корня }

{2}  sreg(1,k,x);{восстановить регулярность 1-поддерева всюду}

{2}  end;

for i:=1 to n do write(f,x[i]:4);

close(f);

end.
Решение 3. Быстрая сортировка Хоара.
В заключение приведем рекурсивный алгоритм сортировки массива, который на практике является одним из самых быстрых. Пусть дан массив a[1]..a[n]. Рекурсивная процедура sort (l,r:integer) сортирует участок массива с индексами из полуинтервала (l,r], то есть a[l+1]..a[r], не затрагивая остального массива.

     procedure sort (l,r: integer);

     begin

     | if (l = r) then begin

     | | ничего делать не надо - участок пуст

     | end else begin

     | | выбрать случайное число s в полуинтервале (l,r]

     | | b := a[s]

     | | переставить элементы сортируемого участка так, чтобы

     | |   сначала шли элементы, меньшие b - участок (l,ll]

     | |   затем элементы, равные b        - участок (ll,rr]

     | |   затем элементы, большие b       - участок (rr,r]

     | | sort (l,ll);

     | | sort (rr,r);

     | end;

     end;

Разделение элементов сортируемого участка на три категории (меньшие, равные, большие) рассматривалось в главе о массивах (это можно сделать за время, пропорциональное длине участка).

Конечность глубины рекурсии гарантируется тем, что длина сортируемого участка на каждом уровне рекурсии уменьшается хотя бы на 1.

procedure sort(l,r: integer);

var

  i,j,x,y: integer;

begin

  i:=l; j:=r; x:=a[(l+r) DIV 2];

  repeat

    while a[i]<x do i:=i+1;

    while x<a[j] do j:=j-1;

    if i<=j then

    begin

      y:=a[i]; a[i]:=a[j]; a[j]:=y;

      i:=i+1; j:=j-1;

    end;

  until i>j;

  if l<j then sort(l,j);

  if i<r then sort(i,r);

end;

Программа может выглядеть так:

{ Turbo Sort }

{ Copyright (c) 1985,90 by Borland International, Inc. }

program qsort;

{$R-,S-}

uses Crt;

{ This program demonstrates the quicksort algorithm, which      }

{ provides an extremely efficient method of sorting arrays in   }

{ memory. The program generates a list of 1000 random numbers   }

{ between 0 and 29999, and then sorts them using the QUICKSORT  }

{ procedure. Finally, the sorted list is output on the screen.  }

{ Note that stack and range checks are turned off (through the  }

{ compiler directive above) to optimize execution speed.        }

const

  max = 20000;

type

  list = array[1..max] of integer;

var

  data: list;

  i: integer;

{ QUICKSORT sorts elements in the array A with indices between  }

{ LO and HI (both inclusive). Note that the QUICKSORT proce-    }

{ dure provides only an "interface" to the program. The actual  }

{ processing takes place in the SORT procedure, which executes  }

{ itself recursively.                                           }

procedure quicksort(var a: list; Lo,Hi: integer);

procedure sort(l,r: integer);

var

  i,j,x,y: integer;

begin

  i:=l; j:=r; x:=a[(l+r) DIV 2];

  repeat

    while a[i]<x do i:=i+1;

    while x<a[j] do j:=j-1;

    if i<=j then

    begin

      y:=a[i]; a[i]:=a[j]; a[j]:=y;

      i:=i+1; j:=j-1;

    end;

  until i>j;

  if l<j then sort(l,j);

  if i<r then sort(i,r);

end;

begin {quicksort};

  sort(Lo,Hi);

end;

begin {qsort}

  assign(output,'output.txt');

  rewrite(output);

{  Write('Now generating 1000 random numbers...');}

  Randomize;

  for i:=1 to max do data[i]:=Random(30000);

{  Writeln;

  Write('Now sorting random numbers...');}

  quicksort(data,1,max);

{  Writeln;}

  for i:=1 to max do Write(data[i]:8);

  close(output);

end.
3. Задачи.
1. Даны две целочисленных таблицы А [1:10] и В[1:15]. Разработать алгоритм и написать программу, которая проверяет, являются ли эти таблицы похожими. Две таблицы называются похожими, если совпадают множества чисел, встречающихся в этих таблицах.
Решение:

Мы можем отсортировать оба массива - и A, и B (например, по неубыванию), далее, если первые элементы массивов A и B совпадают, то ищем и в A, и в B минимальные элементы, большие данного, и повторяем сравнения; если же элементы не совпадают, либо один из массивов уже закончился, а другой еще нет, то массивы не похожие.
     { A и B уже отсортированы }

     i:=1; j:=1;    { смотрим массивы A и B, начиная с первых }

                    { элементов }

     while (i<=10) and (j<=15) and (A[i]=B[j]) do

     begin

       element:=A[i];

       while (i<=10) and (A[i]=element) do

         i:=i+1;    { поиск несовпадающего элемента }

       while (j<=15) and (B[j]=element) do

         j:=j+1;    { поиск несовпадающего элемента }

     end;

     if (i=11) and (j=16)   { просмотрели все элементы A и B }

     then writeln('Массивы похожи')

     else writeln('Массивы непохожи');

2. Задается словарь. Найти в нем все анаграммы (слова, составленные из одних и тех же букв).
Решение:

Каждому слову приписываем его номер в словаре. Сначала сортируем буквы в каждом слове по (например) неубыванию. Получаем какой-то "ключ", который совпадает у всех слов-анаграмм (например, слова 'лом' и 'мол' преобразуются в одни и те же ключи 'лмо').

Далее мы сортируем ключи слов (совместно с приписанными номерами) по неубыванию. Все одинаковые ключи будут размещаться в отсортированной последовательности слов друг за другом. Мы просматриваем полученную последовательность, ищем совпадающие ключи и по приписанным им номерам находим в словаре соответствующие слова-анаграммы.

